Assembly Language: IA-32 Instructions

Goals of this Lecture

- Help you learn how to:
  - Manipulate data of various sizes
  - Leverage more sophisticated addressing modes
  - Use condition codes and jumps to change control flow

  … and thereby …

  - Write more efficient assembly-language programs
  - Understand the relationship to data types and common programming constructs in high-level languages
- Focus is on the assembly-language code
  - Rather than layout of memory for storing data (precept)

Handling Different Data Sizes

Variable Sizes in High-Level Language

- C data types vary in size
  - Character: 1 byte
  - Short, int, and long: ??
  - Float and double: ??
  - Pointers: ??
- Programmer-created types
  - Struct: ??
- Arrays
  - Multiple consecutive elements of some fixed size
  - Where each element could be a struct

Supporting Different Sizes in IA-32

- Three main data sizes
  - Byte (b): 1 byte
  - Word (w): 2 bytes
  - Long (l): 4 bytes
- Separate assembly-language instructions
  - E.g., addb, addw, and addl
- Separate ways to access (parts of) a register
  - E.g., for EAX register: %ah or %al, %ax, and %eax
- Larger sizes (e.g., struct)
  - Manipulated in smaller byte, word, or long units

5

Byte Order in Multi-Byte Entities

- IA-32 is a little endian architecture
  - Least significant byte of multi-byte entity is stored at lowest memory address
  - "Little end goes first"

1003 ↘ 00 … 1000 ↙ 05

The 4-byte int 5 (hex 00 00 00 05) at address 1000:

1000	00000101
1001	00000000
1002	00000000
1003	00000000

- Some other systems use big endian
  - Most significant byte of multi-byte entity is stored at lowest memory address
  - "Big end goes first"

1000 ↘ 00 … 1003 ↙ 05

The 4-byte int 5 (hex 00 00 00 05) at address 1000:

1000	00000000
1001	00000000
1002	00000000
1003	00000101

6

Little Endian Example

```
int main(void) {
  int i=0x003377ff, j;
  unsigned char *p = (unsigned char *) &i;
  for (j=0; j<4; j++)
    printf("Byte %d: %x\n", j, p[j]);
}
```

Output on a little-endian machine:

Byte 0: ff
Byte 1: 77
Byte 2: 33
Byte 3: 0

- Portable?

IA-32 General Purpose Registers

31	15 8	7 0	16-bit	32-bit	Common Use
	AH	AL	AX	EAX	Accumulator
	BH	BL	BX	EBX	Pointer to data
	CH	CL	CX	ECX	Counter for loops
	DH	DL	DX	EDX	I/O pointer
	SI			ESI	Pointers (string
	DI			EDI	source and dest)

General-purpose registers

- EBP: pointer to data on stack
- ESP: stack pointer

C Example: One-Byte Data

Global *char* variable i is in *%al*, the *lower byte* of the "A" register.

```
char i;
…
if (i > 5) {
  i++;
else
  i--;
}
```

→

```
        cmpb $5, %al
        jle else
        incb %al
        jmp endif
else:
        decb %al
endif:
```

9

C Example: Four-Byte Data

Global *int* variable i is in *%eax*, the *full 32 bits* of the "A" register.

```
int i;
…
if (i > 5) {
  i++;
else
  i--;
}
```

→

```
        cmpl $5, %eax
        jle else
        incl %eax
        jmp endif
else:
        decl %eax
endif:
```

10

Memory Addressing Modes

11

Loading and Storing Data

- Processors have many ways to access data
  - Known as "addressing modes"
  - Two simple ways seen in previous examples …
- Immediate addressing
  - Example: movl **$0**, %ecx
  - Initialize register ECX with zero
  - Data (e.g., number "0") embedded in the instruction
- Register addressing
  - Example: movl **%edx**, **%ecx**
  - Copy value in register EDX into register ECX
  - Choice of register(s) embedded in the instruction

12

Accessing Memory

- Variables are stored in memory
  - Global and static local variables in Data or BSS section
  - Dynamically allocated variables in the heap
  - Function parameters and local variables on the stack
- Need to be able to load from and store to memory
  - To copy the data between main memory and registers
  - Or manipulate the data directly in memory
- IA-32 has many different addressing modes
  - Corresponding to common programming constructs
  - E.g., accessing a global variable, dereferencing a pointer, accessing a field in a struct, or indexing an array

13

Direct Addressing

- Useful when the address is known in advance
  - Global variables in the Data or BSS sections
- Load or store from a particular memory location
  - Memory address is embedded in the instruction
  - Instruction reads from or writes to that address
- IA-32 example: movl **2000**, %ecx
  - Four-byte variable located at address 2000
  - Read four bytes starting at address 2000
  - Load the value into the ECX register
- Can use a label for (human) readability
  - E.g., "i" to allow "movl i, %eax"

14

Indirect Addressing

- Useful when address is not known in advance
  - Dereference a pointer, for dynamically allocated data
- Load or store from a previously-computed address
  - Register with the address is embedded in the instruction
  - Instruction reads from or writes to that address
- IA-32 example: movl **(%eax)**, %ecx
  - EAX register stores a 32-bit address (e.g., 2000)
  - Read long-word variable stored at that address
  - Load the value into the ECX register
  - The "(%eax)" essentially dereferences the pointer stored in register %eax

15

Base Pointer Addressing

- Useful when accessing part of a larger variable
  - Specific field within a "struct"
  - E.g., if "age" starts at the 8th byte of "student" record
- Load or store with an offset from a base address
  - movl offset(r1), r2
  - Register r1 stores the base address
  - Fixed offset also embedded in the instruction
  - Instruction computes the address and does access
- IA-32 example: movl 8(%eax), %ecx
  - EAX register stores a 32-bit base address (e.g., 2000)
  - Offset of 8 is added to compute address (e.g., 2008)
  - Load the value into the ECX register

16

Indexed Addressing

- Load/store with offset made of register, multiplier
  - Fixed base address embedded in the instruction
  - Offset = register * constant multiplier
- Useful to iterate through an array (e.g., a[i])
  - Base is the start of the array (i.e., “a”)
  - Register is the index (i.e., “i”)
  - Multiplier is the size of the element (e.g., 4 for “int”)
- IA-32 example: movl 2000(,%eax,4), %ecx
  - Index register EAX (say, with value of 10)
  - Multiplied by a multiplier of 1, 2, 4, or 8 (here, 4)
  - Added to a fixed base of 2000 (to get 2040)

17

Indexed Addressing Example

```
int a[20];   ← global variable
…
int i, sum=0;
for (i=0; i<20; i++)
    sum += a[i];
```

```
        movl $0, %ecx
        movl $0, %ebx
sumloop:
        movl a(,%ecx,4), %eax
        addl %eax, %ebx
        incl %ecx
        cmpl $19, %ecx
        jle sumloop
```

EAX: temporary
EBX: sum
ECX: i

Effective Address: More Generally

Offset = Base + (Index * scale) + displacement

Base	Index	scale	displacement
eax	eax	1	None
ebx	ebx	2	8-bit
ecx	ecx	4	16-bit
edx	edx	8	32-bit
esp	esp		
ebp	ebp		
esi	esi		
edi	edi		

- Displacement `movl foo, %ebx`
- Base `movl (%eax), %ebx`
- Base + displacement `movl foo(%eax), %ebx`
  `movl 1(%eax), %ebx`
- (Index * scale) + displacement `movl (%edx,%eax,4), %ebx`
- Base + (index * scale) + displacement `movl foo(%edx,%eax,4),%ebx`

Data Access Methods: Summary

- Immediate addressing: data stored in the instruction itself
  - **movl $10, %ecx**
- Register addressing: data stored in a register
  - **movl %eax, %ecx**
- Direct addressing: address stored in instruction
  - **movl foo, %ecx**
- Indirect addressing: address stored in a register
  - **movl (%eax), %ecx**
- Base pointer addressing: indirect plus offset
  - **movl 4(%eax), %ecx**
- Indexed addressing: instruction contains base address, and specifies an index register and a multiplier (1, 2, 4, or 8)
  - **movl 2000(,%eax,1), %ecx**
  - **Can also have an additional displacement register**

Condition Codes and Control Flow

21

Control Flow

- Common case
  - Execute code sequentially
  - One instruction after another
- Sometimes need to change control flow
  - If-then-else
  - Loops
  - Switch
- Two key ingredients
  - Testing a condition
  - Selecting what to run next based on result

```
      cmpl $5, %eax
      jle else
      incl %eax
      jmp endif
else:
      decl %eax
endif:
```

Condition Codes

- 1-bit registers set by arithmetic & logic instructions
  - ZF: Zero Flag
  - SF: Sign Flag
  - CF: Carry Flag
  - OF: Overflow Flag
- Example: "addl Src, Dest" ("t = a + b")
  - ZF: set if t == 0
  - SF: set if t < 0
  - CF: set if carry out from most significant bit (unsigned)
    - *Unsigned* overflow
  - OF: set if two's complement overflow
    - (a>0 && b>0 && t<0)
      || (a<0 && b<0 && t>=0)

23

Condition Codes (continued)

- Example: "cmpl Src2,Src1" (compare b,a)
  - Like computing a-b without setting destination
  - ZF: set if a == b
  - SF: set if (a-b) < 0
  - CF: set if carry out from most significant bit (unsigned)
  - OF: set if two's complement overflow
    - (a>0 && b<0 && (a-b)<0) || (a<0 && b>0 && (a-b)>0)
- Flags are *not* set by lea, inc, or dec instructions
  - Hint: this is useful for the extra-credit part of the assembly-language programming assignment

24

Jumps after Comparison (cmpl)

- Equality
  - Equal: je (ZF is set)
  - Not equal: jne (~ZF)
- Below/above (e.g., unsigned arithmetic)
  - Below: jb (CF is set)
  - Above or equal: jae (~CF)
  - Below or equal: jbe (CF I ZF)
  - Above: ja (~(CF I ZF))
- Less/greater (e.g., signed arithmetic)
  - Less: jl (SF ^ OF)
  - Greater or equal: jge (~(SF ^ OF))
  - Less or equal: jle ((SF ^ OF) I ZF)
  - Greater: jg (!((SF ^ OF) I ZF))

26

Branch Instructions

- Conditional jump
  - j{l,g,e,ne,...} target   if (condition) {eip = target}

Comparison	Signed	Unsigned	
=	e	e	*"equal"*
≠	ne	ne	*"not equal"*
>	g	a	*"greater,above"*
≥	ge	ae	*"...-or-equal"*
<	l	b	*"less,below"*
≤	le	be	*"...-or-equal"*
overflow/carry	o	c	
no ovf/carry	no	nc	

- Unconditional jump
  - jmp target
  - jmp *register

27

Jumping

- Simple model of a "goto" statement
  - Go to a particular place in the code
  - Based on whether a condition is true or false
  - Can represent if-the-else, switch, loops, etc.
- Pseudocode example: If-Then-Else

```
if (Test) {
     then-body;
} else {
     else-body;
```

⇒

```
  if (!Test) jump to Else;
  then-body;
  jump to Done;
Else:
  else-body;
Done:
```


Jumping (continued)

- Pseudocode example: Do-While loop

```
do {
  Body;
} while (Test);
```

⇒

```
loop:
  Body;
  if (Test) then jump to loop;
```

- Pseudocode example: While loop

```
while (Test)
  Body;
```

⇒

```
   jump to middle;
loop:
   Body;
middle:
   if (Test) then jump to loop;
```

Jumping (continued)

- Pseudocode example: For loop

```
for (Init; Test; Update)
    Body
```

```
   Init;
   if (!Test) jump to done;
loop:
   Body;
   Update;
   if (Test) jump to loop;
done:
```

30

Example Instruction Types

31

Arithmetic Instructions

- Simple instructions
  - add{b,w,l} source, dest — dest = source + dest
  - sub{b,w,l} source, dest — dest = dest – source
  - Inc{b,w,l} dest — dest = dest + 1
  - dec{b,w,l} dest — dest = dest – 1
  - neg{b,w,l} dest — dest = ~dest + 1
  - cmp{b,w,l} source1, source2 — source2 – source1
- Multiply
  - mul (unsigned) or imul (signed)
- Divide
  - div (unsigned) or idiv (signed)
- Many more in Intel manual (volume 2)
  - adc, sbb, decimal arithmetic instructions

Bitwise Logic Instructions

- Simple instructions

and{b,w,l} source, dest	dest = source & dest
or{b,w,l} source, dest	dest = source | dest
xor{b,w,l} source, dest	dest = source ^ dest
not{b,w,l} dest	dest = ~dest
sal{b,w,l} source, dest (arithmetic)	dest = dest << source
sar{b,w,l} source, dest (arithmetic)	dest = dest >> source

- Many more in Intel Manual (volume 2)
  - Logic shift
  - Rotation shift
  - Bit scan
  - Bit test
  - Byte set on conditions

Data Transfer Instructions

- **`mov{b,w,l} source, dest`**
  - General move instruction
- **`push{w,l} source`**

```
pushl %ebx    # equivalent instructions
                  subl $4, %esp
                  movl %ebx, (%esp)
```



- **`pop{w,l} dest`**

```
popl %ebx     # equivalent instructions
                  movl (%esp), %ebx
                  addl  $4, %esp
```



- Many more in Intel manual (volume 2)
  - Type conversion, conditional move, exchange, compare and exchange, I/O port, string move, etc.

Conclusions

- Accessing data
  - Byte, word, and long-word data types
  - Wide variety of addressing modes
- Control flow
  - Common C control-flow constructs
  - Condition codes and jump instructions
- Manipulating data
  - Arithmetic and logic operations
- Next time
  - Calling functions, using the stack